
options notes nosource; 

proc datasets lib=work nolist memtype=data kill; quit; 

* macro to save output and log to appropriate areas ; 

%_mprintto; 

%put NOTE: 

=======================================================================; 

%put NOTE: Covance Study Number : 000000106326; 

%put NOTE: Client Protocol ID   : ZRHM-PK-05-JP; 

%put NOTE: Program Name         : d_2ADDS.sas; 

%put NOTE: Purpose              : create ADDS dataset; 

%put NOTE: ; 

%put NOTE: Input Data           : STDLIB.ADDS SDTM.DS ADAM.ADSL 

SDTM.SUPPDS; 

%put NOTE: Output               : ADAM.ADDS; 

%put NOTE: Macros Called        : _MPRINTTO _MTOTPER _MPERALL _SCRAMBLE; 

%put NOTE: ; 

%put NOTE: Programmed by        : cvn_jhardman; 

%put NOTE: Creation Date        : 2014-01-03; 

%put NOTE: SAS Version          : 9.3; 

%put NOTE: ; 

%put NOTE: == Latest Run 

=========================================================; 

%put NOTE: Run by               : &sysuserid; 

%put NOTE: Date/Time            : 

%sysfunc(putn(%sysfunc(date()),e8601da.))T%sysfunc(putn(%sysfunc(time()),

e8601lz.)); 

%put NOTE: ; 

%put NOTE: == Modification History 

===============================================; 

%put NOTE: Date        Initials   No. Reason; 

%put NOTE: 22Apr2014   KB         1) Added in OTHER to keep;  

%put NOTE: 05Aug2014   KB         2) Added EXNOTRFL and NICOGR2 to keep;  

%put NOTE: 06Aug2014   KB         3) Amended key variables;  

%put NOTE: ; 

%put NOTE: ; 

%put NOTE: 

=======================================================================; 

options notes source source2 nofullstimer validvarname=upcase missing=' 

'; 

ods _all_ close; 

ods listing; 

 

*=====================================================================; 

* START OF PROGRAM CODE                                               ; 

*=====================================================================; 

****************; 

* bring in ADSL ; 

****************; 

 

data adsl; 

 set adam.adsl; 

 keep studyid usubjid subjid: siteid age sex: race height weightbl 

bmi ucpdgr1 ucpdgr1n nicogr1 nicogr1n NICOGR2 NICOGR2N targr1 targr1n  /* 

2) KB 05Aug2014 */  



   enrlfl scrffl exfl enfl complfl fupfl saffl pprotfl randfl 

trtsdt tr01: tr02:  

   lvisdt lvisday trt01: trt02: dthfl trtseq: analgr1 analgr1n 

EXNOTRFL; /* 2) KB 05Aug2014 */  

run; 

 

************************; 

* bring in SUPPDS ; 

************************; 

 

proc transpose data = sdtm.suppds out=suppds(drop = _:); 

 var qval; 

 by usubjid idvarval; 

 id qnam; 

run; 

 

data suppds2(drop = idvarval); 

 set suppds; 

 dsseq = input(idvarval,best.); 

run; 

 

proc sort data=suppds2; 

 by usubjid dsseq; 

run; 

 

****************; 

* bring in DS   ; 

****************; 

proc sort data = sdtm.ds out = ds; 

 by usubjid dsseq; 

run; 

 

data ds2; 

 merge ds suppds2; 

 by usubjid dsseq; 

 format astdtm datetime13. adt astdt date9. asttm time5.; 

 

 if length(dsstdtc) gt 10 then astdtm = input(dsstdtc,e8601dt.); 

 

 adt = input(dsdtc,yymmdd10.); 

 

 if not missing(astdtm) then astdt = datepart(astdtm); 

 else if length(dsstdtc) = 10 then astdt = input(dsstdtc,yymmdd10.); 

 

 if not missing(astdtm) then asttm = timepart(astdtm); 

 

 keep usubjid dsseq dsterm dsdecod dscat dsdtc dsstdtc dsstdy  adt 

astdtm astdt asttm epoch OTHER; /* 1) KB 22Apr2014 */  

run; 

 

 

***************************; 

* Combine ADSL and DS data *; 

***************************; 



* find periods; 

%_mtotper; 

 

data slds(drop = tr01: tr02: trt01: trt02: trtsdt DSSTDY);   

 merge adsl ds2(in = a); 

 by usubjid; 

 if a;  * only include subjects with DS data ; 

 format aday astday aperiod trtan trtpn 8. trta trtp $40. aperiodc 

$8.; 

 aday = adt - trtsdt + 1; 

 astday = astdt - trtsdt + 1; 

 if astday in (0 1) then aperiod=1; 

 else if astday in (2 3) then aperiod=2; 

 %_mperall(dvar1 = astdtm, dvar2 = astdt); 

 if not missing(aperiod) then do; 

  aperiodc = 'Period ' || put(aperiod,1.); 

 end; 

run; 

 

 

************************; 

* create output dataset ; 

************************; 

 

options replace; 

 

data adds; 

 set stdlib.adds slds;  

 label aperiodc = 'Period (C)'; 

run; 

 

proc sort data = adds out = adam.adds(label= 'Disposition Analysis 

Dataset'); 

/* by usubjid dsstdtc dsdtc dsdecod dsterm;*/ 

    BY USUBJID DSSTDTC DSDECOD DSTERM; /* 3) KB 06Aug2014 */  

run; 

 

options noreplace; 

 

%_scramble(set=adds, id=usubjid subjid subjidn age sex sexc sexn race 

dthfl height weightbl bmi ucpdgr1 ucpdgr1n nicogr1  

   nicogr1n targr1 targr1n analgr1 analgr1n,  

   nullc=trtp trta trtseqp trtseqa,  

   nulln=trtpn trtan trtseqpn trtseqan, nullcc=4, 

nullnc=4); 

 

 

proc printto; run; 

*=====================================================================; 

* END OF PROGRAM CODE                                                 ; 

*=====================================================================; 

 

 

 


